Paper 3053-2019

Using SAS® Procedures to Fit Rasch Models

Tianshu Pan, Pearson; Yumin Chen, Frost Bank

ABSTRACT

This article goes over the SAS® procedures which can fit the Rasch model, such as the IRT, LOGISTIC, GENMOD, GLIMMIX, NLMIXED, NLIN and MCMC procedures, and describes how to use them to fit the standard Rasch model and the Rasch model with all person or item parameters fixed.

INTRODUCTION

Item Response Theory (IRT; Lord, 1980) models are widely used in educational and psychological testing. Usually the IRT models are fitted using special computer software, such as WINSTEPS (Linacre, 2011b), and BILOG-MG (Zimowski, Muraki, Mislevy & Bock, 2003). WINSTEPS implements the joint maximum likelihood (JML; Wright & Douglas, 1977; Wright & Panchapakesan, 1969; Wright & Stone, 1979) and can fit the Rasch model (Rasch, 1960) or Rasch-styled models only while BILOG-MG uses the marginal maximum likelihood (MML) which Bock and Aitkin (1981) proposed and can fit the Rasch model, the two- and three-parameter logistic IRT models.

Besides these IRT software packages, some SAS (SAS Institute, 2017a) procedures can also fit IRT models. For example, the LOGISTIC procedure can use the maximum likelihood (ML) to estimate the standard Rasch model and the parameter estimates are comparable to WINSTEPS's (Pan & Chen, 2011), the IRT procedure can fit IRT models using MML (An & Yung, 2014), and the MCMC procedure can fit IRT models and obtain Bayesian parameter estimates (Stone & Zhu, 2015). Because the GENMOD, GLIMMIX, NLMIXED and NLIN procedures can provide the results equivalent or comparable to what the LOGISTIC procedure gives under some conditions, they can also fit the Rasch model. Specially, the GENMOD procedure with the BAYES statement can provide Bayesian parameter estimates for the Rasch model as well.

Furthermore, in some studies (e.g., Uekawa, 2005; Nord, 2008; Tan, Su & Hogan, 2011) the person parameter of the Rasch was specified as a random effect in the NLMIXED and GLIMMIX procedures to fit the Rasch model, which is similar to what MML does. But their SAS code will not be discussed in this article because the estimation method and algorithms used in the two procedures are different from the description of Bock and Aitkin (1981), and the parameter estimates are not comparable to what the IRT software packages, such as WINSTEPS and BILOG-MG, obtain.

In this article, we will go over these SAS procedures which can fit the Rasch model, illustrate how to use them to fit the standard Rasch model and the Rasch model with all person or item parameters fixed, and compare their parameter estimates with what the IRT software packages BILOG-MG and WINSTEPS give.

USING SAS PROCEDURES TO FIT THE STANDARD RASCH MODEL

A standard Rasch model could be specified as:

$$\Pr(y_{ij}=1)=\frac{\exp(\theta_i-b_j)}{1+\exp(\theta_i-b_j)} ,$$

where y_{ij} is the score of person i on item j, person parameter θ_i is the ability of person i, and item parameter b_j is the difficulty of item j.

To show how to use SAS procedures to fit the Rasch model mentioned above, we first created a simulated test data set with nine items and 20 persons or examinees. The data can be input using the following SAS code:

```
DATA data1;
  INPUT id $ s1-s9;
CARDS;
01 1 0 1 0 0 0 0 0 0
02 1 0 0 1 0 0 0 0 0
…
;
```

In the data set, the nine variables 's1' – 's9' are item scores, and the variable 'id' is the ID numbers of persons. Then it is easy to use the IRT procedure to fit a standard Rasch model:

```
PROC IRT DATA=data1 RESFUNC=RASCH QPOINTS=32 ABSFCONV=0.0001 TECH=EM
         SCOREMETHOD=ML OUT=out1;
  VAR s1-s9;
RUN;
```

In the code above, the options, QPOINTS, ABSFCONV and TECH, are used to estimate item parameters, and the SCOREMETHOD and OUT options can help get the person parameter estimates. We used these options intentionally so that its results can be compared with what BILOG-MG obtained under the same conditions. After the commands of BILOG-MG were adjusted accordingly, by Table 1, the differences were only 0.017 between their correspondent parameter estimates. Thus, their results were comparable.

Pan (2011; 2018) argued that under ML the standard Rasch model is equivalent to a special binary logistic regression model in which the intercept is not specified, item scores are the dependent variable, items are effect-coded, and all persons are dummy-coded to be the independent variables. Then, any SAS procedure fitting logistic regression models can fit the standard Rasch model, such as the LOGISTIC, GENMOD, GLIMMIX, NLMIXED or NLIN procedure. The LOGISTIC procedure fits logistic regression models. Although usually the GENMOD and GLIMMIX procedures fit generalized linear models (GLM) and generalized linear mixed models (GLMM) respectively, they can also fit a logistic regression model because logistic regression models are a type of GLM and can be treated as a special case of GLMM, i.e., GLMM without G-side random effects. Moreover, a logistic regression model can also be specified using the flexible programming statements in the NLMIXED and NLIN procedure.

Because a binary logistic regression model has only one dependent variable, the simulated data abovementioned cannot be used in these SAS procedures. The data need to be transposed to put all item scores in one variable or input using the following code:

```
DATA data2;
  INPUT score total item id $;
CARDS;
1 2 1 01
0 2 2 01
…
0 6 8 20
0 6 9 20
;
```

This SAS data set, 'data2', has the following variables: 'item' is the item sequence numbers, 'id' is the ID numbers of examinees, 'score' is examinees' item scores, and 'total' is examinees' total raw scores on the test.

Table 1. Comparisons of Parameter Estimates Between SAS Procedures, BILOG-MG and WINSTEPS

Parameters	BILOG-MG	WINSTEPS	SAS Procedures					
			IRT	LOGISTIC GENMOD GLIMMIX NLMIXED	NLIN	LOGISTIC with FIRTH	GENMOD with BAYES	MCMC
Items	Item Parameter Estimates							
1	-1.7162	-1.9730	-1.7335	-1.9731	-1.9729	-1.6062	-1.8122	-1.7757
2	-0.7706	-0.9028	-0.7879	-0.9028	-0.9027	-0.7415	-0.8834	-0.7693
3	-0.5001	-0.5854	-0.5174	-0.5855	-0.5854	-0.4807	-0.5253	-0.4881
4	-1.3674	-1.5866	-1.3847	-1.5867	-1.5866	-1.2974	-1.4706	-1.3921
5	-0.2389	-0.2771	-0.2562	-0.2771	-0.2771	-0.2262	-0.2861	-0.1783
6	0.5352	0.6327	0.5179	0.6328	0.6328	0.5259	0.5625	0.5680
7	0.8054	0.9457	0.7881	0.9458	0.9457	0.7828	0.8484	0.8113
8	1.0911	1.2736	1.0738	1.2737	1.2736	1.0497	1.1761	1.0484
9	2.1614	2.4729	2.1441	2.4729*	2.4726*	1.9936*	2.3906*	2.1341
Total Scores	Person Parameter Estimates							
1	-2.5648	-2.6852	-2.5475	-2.6853	-2.6850	-2.1170	-1.4253	–
2	-1.6199	-1.7078	-1.6026	-1.7079	-1.7078	-1.3920	-1.4984	–
3	-0.9353	-0.9870	-0.9180	-0.9870	-0.9869	-0.8121	-0.5629	–
4	-0.3337	-0.3458	-0.3164	-0.3459	-0.3458	-0.2837	-0.2682	–
5	0.2506	0.2808	0.2679	0.2808	0.2808	0.2354	0.2125	–
6	0.2506	0.2808	0.2679	0.2808	0.9397	0.7770	0.7977	–
7	0.8656	0.9397	0.8829	0.9398	1.6988	1.3837	1.2512	–
8	2.5717	2.7411	2.5891	2.7412	2.7407	2.1512	1.8101	–

Note: * indicates the parameter estimate was obtained through the sum-to-zero constraint.

Then, in terms of Pan and Chen (2011), we need 20 indicator variables for 20 persons or examinees and eight effect-coded variables for nine items to fit the standard Rasch model. Thus, we created the effect-coded variables 'i1' – 'i8' corresponding to Items 1 – 8 respectively, and the last item, Item 9, was used as the reference. If the standard effect-coding is used, then the item parameter estimates obtained using the LOGISTIC procedure will have the opposite signs to the estimates from WINSTEPS as what Pan and Chen (2011) got. To avoid it, we reversed the signs of numbers used in the standard effect coding when creating the coded variables 'i1' – 'i8' for items, i.e., the values of a coded variable for an item were equal to −1 when a score was from this item, 1 when a score was from the reference item, and 0 when a score was from the other items. This coding still put the sum-to-zero constraint on the item parameters. Then, the below LOGISTIC code can be used to obtain item parameter estimates comparable to WINSTEPS's without any transformation.

```
PROC LOGISTIC DATA=data2 DESCENDING;
  CLASS id / PARAM=GLM;
  MODEL score = id i1-i8 / NOINT TECH=NEWTON ABSFCONV=0.0001;
RUN;
```

In the code above, the DESCENDING option will reverse the sort order for the levels of the response variable when the LOGISTIC procedure does modeling, and here it forces this procedure to model the probability that 'score' = 1 as the Rasch model does; the NOINT option is used to fit a model without an intercept as the model has no intercept; the option PARAM=GLM forces the procedure to dummy-code every class level of the classification variable specified in the CLASS statement and create the indicator variables needed; for the purpose of the comparison, the options TECH and ABSFCONV specify the optimization algorithm (Newton-Raphson) and convergence criterion (0.0001) respectively. Accordingly, the code of other SAS procedures mentioned later and the WINSTEPS commands were adjusted too.

Because the person parameter estimates are the same for examinees taking the same item set and receiving the same total score in the Rasch model, the variable 'total' can replace the variable 'id' in the above SAS code, and the code can run much faster if there are hundreds of examinees (Pan & Chen, 2011). Then, the code can be revised as follows:

```
PROC LOGISTIC DATA = data2 DESCENDING;
  CLASS total / PARAM=GLM;
  MODEL score = total i1-i8 / NOINT TECH=NEWTON ABSFCONV=0.0001;
RUN;
```

Table 1 shows that the differences between parameter estimates of the LOGISTIC procedure and WINSTEPS was only 0.0001 as Pan (2011; 2018) obtained.

When the GENMOD and GLIMMIX procedures were used, the below SAS code gave the same parameter estimates as the above LOGISTIC code obtained.

```
PROC GENMOD DATA=data2 DESCENDING;
  CLASS total;
  MODEL score = total i1-i8 / NOINT DIST=BIN LINK=LOGIT CONVERGE=0.0001;
RUN;
PROC GLIMMIX DATA=data2;
  CLASS total;
  MODEL score (DESCENDING) = total i1-i8 / NOINT DIST=BINARY LINK=LOGIT S;
  NLOPTIONS TECH=NEWRAP ABSFCONV=0.0001;
RUN;
```

The option PARAM=GLM is not needed in the above code because the GENMOD procedure uses it by default, and the GLIMMIX procedure uses this coding only; the options DIST and LINK are used to specify a logistic regression model.

If the NLMIXED and NLIN procedures are applied, besides the coded variables 'i1' – 'i8' for items, the indicator variables for all different total scores need to be created as the two procedures do not have the CLASS statement. Because the values of total scores were 1 to 8 in the simulated data as Table 1 shows, we needed to create eight indicator variables, 't1' – 't8', which are correspondent to eight total scores respectively.

The below NLMIXED code gave the same estimates as the LOGISTIC, GENMOD, and GLIMMIX procedures obtained:

```
PROC NLMIXED DATA=data2 TECH=NEWRAP ABSFCONV=0.0001 NOAD;
  PARMS a1=0 a2=0 a3=0 a4=0 a5=0 a6=0 a7=0 a8=0
        b1=0 b2=0 b3=0 b4=0 b5=0 b6=0 b7=0 b8=0;
  eta = a1*t1+a2*t2+a3*t3+a4*t4+a5*t5+a6*t6+a7*t7+a8*t8
       +b1*i1+b2*i2+b3*i3+b4*i4+b5*i5+b6*i6+b7*i7+b8*i8;
  pi = EXP(eta)/(1+EXP(eta));
  MODEL score ~ BINARY(pi);
RUN;
```

In the above code, the PARMS statement provides starting values of all parameters specified, and defines points in the parameter spaces of the model specified; the MODEL statement specifies that the variable 'score' follows a binary (Bernoulli) distribution with the probability 'pi'; In the programming statements, 'eta' creates the linear predictors of a logistic regression model, and 'pi' is the logistic transformation of 'eta'; the coefficients 'a1'

– ‘a8’ represent the person parameters with different total raw scores respectively; the coefficients ‘b1’ – ‘b8’ represent the parameters of the first eight items respectively, and the ninth-item parameter can still be obtained through the sum-to-zero constraint.

It is a little complicated to use the NLIN procedure and fit the Rasch model. SAS/STAT® user’s guide (SAS Institute, 2017b) provides a NLIN code example to fit a probit model by minimizing twice the negative of the log-likelihood function. Based on the example, we first created a ‘dummy’ response variable ‘lk’, whose values are equal to zeros for all observations. Then, the similar programming statements can be used to specify the model, but ‘pi’ should be revised to provide different transformations of ‘eta’ when the variable ‘score’ is equal to 0 and 1, and the MODEL statement specifies twice the negative of the log-likelihood function. The NLIN code is shown as below:

```
PROC NLIN DATA=data2 METHOD=NEWTON CONVERGE=0.0001;
  PARMS a1=0 a2=0 a3=0 a4=0 a5=0 a6=0 a7=0 a8=0
        b1=0 b2=0 b3=0 b4=0 b5=0 b6=0 b7=0 b8=0;
  eta = a1*t1+a2*t2+a3*t3+a4*t4+a5*t5+a6*t6+a7*t7+a8*t8
       +b1*i1+b2*i2+b3*i3+b4*i4+b5*i5+b6*i6+b7*i7+b8*i8;
  pi = EXP(eta*score)/(1+EXP(eta));
  MODEL lk = SQRT(-2*LOG(pi));
RUN;
```

Table 1 shows that its parameter estimates were very close to what was obtained using the LOGISTIC, GENMOD, GLIMMIX and NLMIXED procedures.

However, ML or JML estimates for the Rasch model are biased (Ghosh, 1995, Wright & Douglas, 1977). To reduce the bias, Pan (2011; 2018) suggested using the penalized maximum likelihood (PML; Firth, 1993). With the FIRTH option, the following LOGISTIC code uses PML to fit the standard Rasch model, and the results are shown in Table 1.

```
PROC LOGISTIC DATA = data2 DESCENDING;
  CLASS total / PARAM=GLM;
  MODEL score = total i1-i8 / NOINT FIRTH TECH=NEWTON ABSFCONV=0.0001;
RUN;
```

Another choice is the Bayesian method. The below GENMOD code with the BAYES statement obtains full Bayesian estimates for the Rasch model. In the code, a data set is first created with the means and variances of the prior distributions Stone and Zhu (2014) used.

```
DATA priors;
  INPUT _type_ $ total1-total8 i1-i8;
CARDS;
Mean 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
Var 1 1 1 1 1 1 1 1 25 25 25 25 25 25 25 25
;
PROC GENMOD DATA=data2 DESCENDING;
  CLASS total;
  MODEL score = total i1-i8 / NOINT DIST=BIN LINK=LOGIT;
  BAYES COEFFPRIOR=NORMAL(INPUT=priors) NBI=2000 NMC=10000 SEED=1;
RUN;
```

In addition, the MCMC procedure can also provide Bayesian estimates for the IRT models (Stone & Zhu, 2014). The code below can obtain Bayesian estimates for the Rasch model:

```
PROC MCMC DATA=data1 NBI=2000 NMC=10000 SEED=1;
  ARRAY b[9]; ARRAY s[9]; ARRAY p[9];
  PARMS b: 0;
  PRIOR b: ~ NORMAL(0, VAR=25) ;
  RANDOM theta ~ NORMAL(0, VAR=1) SUBJECT=id;
  DO j=1 TO 9;
    p[j] = 1/(1+exp(-(theta - b[j])));
```

```
    END;
    MODEL s1 ~ BINARY(p1);
    MODEL s2 ~ BINARY(p2);
    MODEL s3 ~ BINARY(p3);
    MODEL s4 ~ BINARY(p4);
    MODEL s5 ~ BINARY(p5);
    MODEL s6 ~ BINARY(p6);
    MODEL s7 ~ BINARY(p7);
    MODEL s8 ~ BINARY(p8);
    MODEL s9 ~ BINARY(p9);
  RUN;
```

As Table 1 shows, the Bayesian results of the MCMC and GENMOD procedure are different but close although their input data have different structures. The differences of item parameter estimates may be due to different sampling methods and random errors. Comparatively, the GENMOD code is much simpler than the MCMC code, and the GENMOD procedure gives Bayesian estimates for both item and person parameters but the MCMC code of Stone and Zhu (2014) only gives Bayesian estimates of item parameters. However, a data set is needed to input the means and variances of the customized prior normal distributions in the GENMOD procedure.

USING SAS TO FIT THE RASCH MODEL WITH ALL PERSON OR ITEM PARAMETERS FIXED

In the previous section, we discussed how to use SAS to estimate the standard Rasch model. However, sometimes anchored items or persons are used to do equating, scaling, or even differential item functioning (DIF) analysis in practice. The IRT programs, e.g., WINSTEPS and BILOG-MG, can do calibration with fixed parameters easily. When WINSTEPS does DIF analysis, it does the joint run of all persons to obtain item and person parameters first, and then each group is run with person parameters anchored at their joint-run values to produce item parameter estimates of every group (Linacre, 2011a). How do we do it in these SAS procedures? We can specify a Rasch model with all person parameters fixed in these SAS procedures if the fixed values of the person parameters are in the data ‘data2’.

Let’s assume that the 20 persons in the simulated data set belong to two groups, i.e., Groups 1 and 2, which have odd- and even-numbered ID numbers respectively. To compare the results with WINSTEPS’s, we merged the person parameter estimates gotten from WINSTEPS with the simulated data and named this new variable ‘theta’. In this case, the sum-to-zero constraint is not used, but it is needed to create the nine indicator variables, ‘item1’ – ‘item9’, for the nine items. Usually the indicator variables are created using the dummy codes 0 and 1. However, we used −1 to replace 1 and create the nine variables, which can cause the estimates to have the same sign as what WINSTEPS gives. Then when the OFFSET option and the new variable ‘theta’ are used with the LOGISTIC, GENMOD and GLIMMIX procedures, we can obtain item parameter estimates with all person parameters fixed. Their parameter estimates were still the same, and very close to what WINSTEPS obtained. The code is shown as below, and the results are in Table 2.

```
  PROC LOGISTIC DATA=data2 DESCENDING;
    WHERE group=1;
    MODEL score = item1-item9 / NOINT TECH=NEWTON ABSFCONV=0.0001
                                OFFSET=theta;
  RUN;
  PROC GENMOD DATA=data2 DESCENDING;
    WHERE group=2;
    MODEL score = item1-item9 / NOINT DIST=BIN LINK=LOGIT CONVERGE=0.0001
                                OFFSET=theta;
  RUN;
```

Table 2. Comparisons of Item Parameter Estimates Between SAS Procedures and WINSTEPS by Groups

Items	WINSTEPS	LOGISTIC GENMOD GLIMMIX NLMIXED	NLIN
Group 1			
1	-1.8493	-1.8493	-1.8493
2	-0.5033	-0.5033	-0.5033
3	-1.1364	-1.1363	-1.1363
4	-1.1364	-1.1363	-1.1363
5	-0.5033	-0.5033	-0.5033
6	0.7349	0.7349	0.7349
7	0.7349	0.7349	0.7349
8	1.4227	1.4227	1.4227
9	2.2347	2.2347	2.2347
Group 2			
1	-2.1003	-2.1003	-2.1003
2	-1.3365	-1.3365	-1.3365
3	-0.0535	-0.0535	-0.0535
4	-2.1003	-2.1003	-2.1003
5	-0.0535	-0.0535	-0.0535
6	0.5390	0.5390	0.5390
7	1.1431	1.1431	1.1431
8	1.1431	1.1431	1.1431
9	2.7561	2.7571	2.7564

```
PROC GLIMMIX DATA=data2;
  WHERE group=2;
  MODEL score(DESCENDING) = item1-item9 / NOINT DIST=BINARY LINK=LOGIT S
                                          OFFSET=theta;
  NLOPTIONS TECH=NEWRAP ABSFCONV=0.0001;
RUN;
```

In the code above, the WHERE statement is used with these SAS procedures so that each group can be analyzed separately.

These coded variables ‘item1’ – ‘item9’ and the variable ‘theta’ can also be used to specify the Rasch model with all person parameters fixed in the NLMIXED and NLIN procedures. In the NLMIXED and NLIN code below, the coefficients ‘b1’ – ‘b9’ represent the parameters of the nine items respectively. Table 2 shows their estimates and compares them with the ones of WINSTEPS DIF analysis.

```
PROC NLMIXED DATA=data2 TECH=NEWRAP ABSFCONV=0.0001 NOAD;
  WHERE group=1;
  PARMS b1=0 b2=0 b3=0 b4=0 b5=0 b6=0 b7=0 b8=0 b9=0;
  eta = theta+b1*item1+b2*item2+b3*item3+b4*item4
       +b5*item5+b6*item6+b7*item7+b8*item8+b9*item9;
  pi = EXP(eta)/(1+EXP(eta));
  MODEL score ~ BINARY(pi);
RUN;
```

```
PROC NLIN DATA=data2 METHOD=NEWTON CONVERGE=0.0001;
  WHERE group=2;
  PARMS b1=0 b2=0 b3=0 b4=0 b5=0 b6=0 b7=0 b8=0 b9=0;
  eta = theta+b1*item1+b2*item2+b3*item3+b4*item4
       +b5*item5+b6*item6+b7*item7+b8*item8+b9*item9;
  pi  = EXP(eta*score)/(1+EXP(eta));
  MODEL lk = SQRT(-2*LOG(pi));
RUN;
```

By Table 2, we can find that the item parameter estimates of the NLIN procedure were the same as the other four procedures obtained except for the last item of Group 2, and the difference is smaller than 0.01. These results were also comparable to What WINSTEPS gave.

Obviously, these SAS code examples can be revised to estimate the person parameters when all item parameters are given. What we need to do is to replace the person ability with an item difficulty parameter and the indicator variables for items with the ones for persons or total scores, and then these procedures can also fit the Rasch model with all item parameters fixed.

CONCLUSION

This article goes over the SAS procedures which can fit the Rasch model and shows that these procedures can provide the comparable results to what BILOG-MG and WINSTEPS give. However, as Pan and Chen (2011) mentioned, two potential problems need to be noted when these procedures using:

At first, when a person or an item has a zero or full responses, i.e., a person answers all test items (in)correctly or all persons answer an item (in)correctly, ML may fail to converge. In WINSTEPS, data are checked and then these persons or items are excluded. Therefore, they should also be excluded when these SAS procedures are used. However, when the LOGISTIC procedure is applied, PML (Firth, 1993) can solve this issue (Pan, 2011; 2018). It is implemented in the LOGISTIC procedure using the MODEL statement option FIRTH as mentioned before. Heinze and Schemper (2002) showed that the method of Firth's always yields finite estimates of parameters under complete or quasi-complete separation. The parameters of the persons and items with zero or full responses can be estimated using PML. If the GENMOD procedure is applied, the BAYES statement may need to be used because it can deal with these persons or items too.

Second, when examinees take different item sets in a test, persons receiving the same total raw score but taking different item sets will have different parameter estimates in the Rasch model. It happens when omitted responses are treated as missing. For example, two examinees receive 39 points on a 40-item test, but one gets an item wrong and the other skips an item, and then their parameter estimates should be different in the Rasch model. Thus, the total score is inappropriate to be used in these SAS procedures under this condition. Pan and Chen (2011) suggested creating a new variable to differentiate examinees by their total scores and the item sets they take and using the new variable to replace the variable 'total' in these SAS procedures.

Generally, it is convenient for SAS users to estimate the Rasch model using the LOGISTIC, GENMOD, and GLIMMIX procedures, and then they can understand the Rasch model and the relationship between the Rasch model and logistic regression better. These SAS procedures, however, are not able to replace WINSTEPS. The standard errors of parameter estimates provided by these SAS procedures and WINSTEPS are different because logistic regression and the Rasch model use different approaches to calculate the standard errors of parameter estimates (Pan, 2011; 2018). These SAS procedures also do not provide fit statistics for items or persons as WINSTEPS does. SAS users would need to write their own SAS code to calculate these fit statistics or use the code of Nord (2008) when they use these SAS procedures to fit the standard Rasch model.

Additionally, it seems that the NLIN and NLMIXED procedures can also fit the Rasch model with some person or item parameters fixed. However, the parameter estimates are completely different from WINSTEPS's under this condition. The reason may be that they implement different computation algorithms. But further studies are needed.

REFERENCES

An, X., & Yung, Y. F. (2014). *Item response theory: What it is and how you can use the IRT procedure to apply it*. Paper presented at SAS Global Forum.

Bock, R. D., & Aitkin, M. (1981). Marginal maximum likelihood estimation of item parameters: Application of an EM algorithm. *Psychometrika*, *46*, 443-459. doi: 10.1007/BF02293801

Firth, D. (1993). Bias reduction of maximum likelihood estimates. *Biometrika*, *80*, 27-38. doi: 10.1093/biomet/80.1.27

Ghosh, M. (1995). Inconsistent maximum likelihood for the Rasch model. *Statistics & Probability Letters*, *23*, 165–170. doi: 10.1016/0167-7152(94)00109-L

Heinze, G., & Schemper, M. (2002). A solution to the problem of separation in logistic regression. *Statistics in Medicine*, *21*, 2409-2419. doi: 10.1002/sim.1047

Linacre, J. M. (2011a). *A user's guide to WINSTEPS, MINISTEP, Rasch-model computer programs*. Chicago: Winsteps.com.

Linacre, J. M. (2011b). *WINSTEPS Rasch measurement* (Version 3.72.2) [Computer program]. Chicago: Winsteps.com.

Lord, F.M. (1980). *Applications of Item Response Theory to practical testing problems*. Hillsdale, NJ: Lawrence Erlbaum.

Nord, M. (2008). *Using SAS to assess the fit of survey response data to the Rasch measurement model*. Presentation at Special Interest Group in Statistics for the Capital PC User Group and the Washington Institute for Operations Research and the Management Sciences, Washington DC. Retrieved Dec. 1, 2010 from http://www.cpcug.org/

Pan, T. (2011). *Estimating the Rasch model as a logistic regression model to reduce bias of maximum likelihood estimate*. Paper presented at the annual meeting of National Council of Measurement in Education, New Orleans, LA.

Pan. T. (2018). Fitting the Rasch model under the logistic regression framework to reduce estimation bias. *Journal of Modern Applied Statistical Methods*, *17*, 16. doi: 10.22237/jmasm/1530028025

Pan, T., & Chen, Y. (2011). *Using PROC LOGISTIC to estimate the Rasch model*. Paper presented at the SAS Global Forum, Las Vegas, NV.

Rasch, G. (1960). *Probabilistic models for some intelligence and attainment tests*. Copenhagen: Danmarks Paedagogiske Institut.

SAS Institute. (2017a). *SAS/STAT® 14.3* [Computer program]. Cary, NC: SAS Institute.

SAS Institute. (2017b). *SAS/STAT® 14.3 user's guide*. Cary, NC: SAS Institute.

Stone, C. A., & Zhu, X. (2015). *Bayesian analysis of item response theory models using SAS*. Cary, NC: SAS Institute.

Tan, T., Su, S., & Hogan, D. (2011) *Rasch model and beyond with PROC GLIMMIX: A better choice*. Paper presented at the SAS Global Forum, Las Vegas, NV.

Uekawa, K. (2005). *How to Do Rasch Model Using SAS*. Retrieved Feb. 16, 2010 from http://www.estat.us

Wright, B. D., & Douglas, G.A. (1977). Best procedures for sample-free item analysis. *Applied Psychological Measurement*, *1*, 281-294. doi: 10.1177/014662167700100216

Wright, B. D., & Panchapakesan, N. (1969). A procedure for sample-free item analysis. *Educational and Psychological Measurement*, *29*, 23-48. doi: 10.1177/014662167700100216

Wright, B. D., & Stone, M. H. (1979). *Best test design*. Chicago: Measurement, Evaluation, Statistics, and Assessment Press.

Zimowski, M., Muraki, E., Mislevy, R. J., & Bock, R. D. (2003). *BILOG-MG 3: Item analysis and test scoring with binary logistic models* [Computer software]. Chicago, IL: Scientific Software.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Tianshu Pan
Pearson
tianshu.pan@pearson.com